// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: contracts/MultiProposerableTransactionExecutor.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.17;

import {Ownable} from "@openzeppelin/contracts/access/Ownable.sol";

interface ITransferOwnership {
    function transferOwnership(address _owner) external;
}

contract MultiProposerableTransactionExecutor is Ownable {
    event ProposeTransaction(
        address indexed transactionProposer,
        uint256 indexed txIndex,
        address indexed to,
        bytes data
    );
    event ExecuteTransaction(address indexed owner, uint256 indexed txIndex);
    event ExecuteTransactionFailed(
        address indexed owner,
        uint256 indexed txIndex
    );
    event TransferTargetOwnership(address target, address owner);
    event AddTransactionProposer(address transactionProposer);
    event RemoveTransactionProposer(address transactionProposer);

    address[] public transactionProposers;
    mapping(address => bool) public isTransactionProposer;

    struct Transaction {
        address to;
        bytes data;
        bool executed;
        bool failed;
    }

    Transaction[] public transactions;

    function addTransactionProposer(
        address _transactionProposer
    ) external onlyOwner {
        require(
            _transactionProposer != address(0) &&
                isTransactionProposer[_transactionProposer] == false &&
                owner() != _transactionProposer,
            "is already exist in transactionProposers or is the owner"
        );
        isTransactionProposer[_transactionProposer] = true;
        transactionProposers.push(_transactionProposer);

        emit AddTransactionProposer(_transactionProposer);
    }

    function removeTransactionProposer(
        address _transactionProposer
    ) external onlyOwner {
        require(
            isTransactionProposer[_transactionProposer] == true,
            "is not exist in transactionProposers"
        );
        isTransactionProposer[_transactionProposer] = false;

        for (uint256 i = 0; i < transactionProposers.length - 1; i++) {
            if (transactionProposers[i] == _transactionProposer) {
                transactionProposers[i] = transactionProposers[
                    transactionProposers.length - 1
                ];
                break;
            }
        }
        transactionProposers.pop();

        emit RemoveTransactionProposer(_transactionProposer);
    }

    function proposeTransaction(address _to, bytes memory _data) external {
        require(
            owner() == msg.sender || isTransactionProposer[msg.sender],
            "not owner or not transactionProposer"
        );

        require(_to != address(0), "_to can't be zero address");
        require(_data.length != 0, "_data must be exist");

        uint256 txIndex = transactions.length;

        transactions.push(
            Transaction({to: _to, data: _data, executed: false, failed: false})
        );

        emit ProposeTransaction(msg.sender, txIndex, _to, _data);
    }

    function executeTransaction(uint256 _txIndex) external onlyOwner {
        require(_txIndex < transactions.length, "tx does not exist");
        require(!transactions[_txIndex].executed, "tx already executed");

        Transaction storage transaction = transactions[_txIndex];

        transaction.executed = true;

        (bool success, ) = transaction.to.call(transaction.data);
        if (success) {
            emit ExecuteTransaction(msg.sender, _txIndex);
        } else {
            transaction.failed = true;
            emit ExecuteTransactionFailed(msg.sender, _txIndex);
        }
    }

    function transferTargetOwnership(
        address _target,
        address _owner
    ) external onlyOwner {
        require(_owner != address(0), "_owner can't be 0");

        ITransferOwnership target = ITransferOwnership(_target);

        target.transferOwnership(_owner);

        emit TransferTargetOwnership(_target, _owner);
    }

    function getTransactionProposers()
        external
        view
        returns (address[] memory)
    {
        return transactionProposers;
    }

    function getTransactionCount() external view returns (uint256) {
        return transactions.length;
    }

    function getTransaction(
        uint256 _txIndex
    )
        external
        view
        returns (address to, bytes memory data, bool executed, bool failed)
    {
        Transaction storage transaction = transactions[_txIndex];

        return (
            transaction.to,
            transaction.data,
            transaction.executed,
            transaction.failed
        );
    }
}